

NAG Library Routine Document

F07AUF (ZGECON)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F07AUF (ZGECON) estimates the condition number of a complex matrix A , where A has been factorized by F07ARF (ZGETRF).

2 Specification

```
SUBROUTINE F07AUF (NORM, N, A, LDA, ANORM, RCOND, WORK, RWORK, INFO)
INTEGER                N, LDA, INFO
REAL (KIND=nag_wp)    ANORM, RCOND, RWORK(2*N)
COMPLEX (KIND=nag_wp) A(LDA,*), WORK(2*N)
CHARACTER(1)          NORM
```

The routine may be called by its LAPACK name *zgecon*.

3 Description

F07AUF (ZGECON) estimates the condition number of a complex matrix A , in either the 1-norm or the ∞ -norm:

$$\kappa_1(A) = \|A\|_1 \|A^{-1}\|_1 \quad \text{or} \quad \kappa_\infty(A) = \|A\|_\infty \|A^{-1}\|_\infty.$$

Note that $\kappa_\infty(A) = \kappa_1(A^H)$.

Because the condition number is infinite if A is singular, the routine actually returns an estimate of the **reciprocal** of the condition number.

The routine should be preceded by a call to F06UAF to compute $\|A\|_1$ or $\|A\|_\infty$, and a call to F07ARF (ZGETRF) to compute the LU factorization of A . The routine then uses Higham's implementation of Hager's method (see Higham (1988)) to estimate $\|A^{-1}\|_1$ or $\|A^{-1}\|_\infty$.

4 References

Higham N J (1988) FORTRAN codes for estimating the one-norm of a real or complex matrix, with applications to condition estimation *ACM Trans. Math. Software* **14** 381–396

5 Arguments

1: NORM – CHARACTER(1) *Input*

On entry: indicates whether $\kappa_1(A)$ or $\kappa_\infty(A)$ is estimated.

NORM = '1' or 'O'

$\kappa_1(A)$ is estimated.

NORM = 'I'

$\kappa_\infty(A)$ is estimated.

Constraint: NORM = '1', 'O' or 'I'.

- 2: N – INTEGER *Input*
On entry: n , the order of the matrix A .
Constraint: $N \geq 0$.
- 3: A(LDA,*) – COMPLEX (KIND=nag_wp) array *Input*
Note: the second dimension of the array A must be at least $\max(1, N)$.
On entry: the LU factorization of A , as returned by F07ARF (ZGETRF).
- 4: LDA – INTEGER *Input*
On entry: the first dimension of the array A as declared in the (sub)program from which F07AUF (ZGECOM) is called.
Constraint: $LDA \geq \max(1, N)$.
- 5: ANORM – REAL (KIND=nag_wp) *Input*
On entry: if NORM = '1' or 'O', the 1-norm of the **original** matrix A .
 If NORM = 'I', the ∞ -norm of the **original** matrix A .
 ANORM may be computed by calling F06UAF with the same value for the argument NORM.
 ANORM must be computed either **before** calling F07ARF (ZGETRF) or else from a **copy** of the original matrix A (see Section 10).
Constraint: $ANORM \geq 0.0$.
- 6: RCOND – REAL (KIND=nag_wp) *Output*
On exit: an estimate of the reciprocal of the condition number of A . RCOND is set to zero if exact singularity is detected or the estimate underflows. If RCOND is less than **machine precision**, A is singular to working precision.
- 7: WORK($2 \times N$) – COMPLEX (KIND=nag_wp) array *Workspace*
- 8: RWORK($2 \times N$) – REAL (KIND=nag_wp) array *Workspace*
- 9: INFO – INTEGER *Output*
On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = $-i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed estimate RCOND is never less than the true value ρ , and in practice is nearly always less than 10ρ , although examples can be constructed where RCOND is much larger.

8 Parallelism and Performance

F07AUF (ZGECOM) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

A call to F07AUF (ZGECOM) involves solving a number of systems of linear equations of the form $Ax = b$ or $A^Hx = b$; the number is usually 5 and never more than 11. Each solution involves approximately $8n^2$ real floating-point operations but takes considerably longer than a call to F07ASF (ZGETRS) with one right-hand side, because extra care is taken to avoid overflow when A is approximately singular.

The real analogue of this routine is F07AGF (DGECON).

10 Example

This example estimates the condition number in the 1-norm of the matrix A , where

$$A = \begin{pmatrix} -1.34 + 2.55i & 0.28 + 3.17i & -6.39 - 2.20i & 0.72 - 0.92i \\ -0.17 - 1.41i & 3.31 - 0.15i & -0.15 + 1.34i & 1.29 + 1.38i \\ -3.29 - 2.39i & -1.91 + 4.42i & -0.14 - 1.35i & 1.72 + 1.35i \\ 2.41 + 0.39i & -0.56 + 1.47i & -0.83 - 0.69i & -1.96 + 0.67i \end{pmatrix}.$$

Here A is nonsymmetric and must first be factorized by F07ARF (ZGETRF). The true condition number in the 1-norm is 231.86.

10.1 Program Text

```

Program f07aufe

!      F07AUF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: nag_wp, x02ajf, zgecon, zgetrf, zlange => f06uaf
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
      Character (1), Parameter   :: norm = '1'
!      .. Local Scalars ..
      Real (Kind=nag_wp)         :: anorm, rcond
      Integer                     :: i, info, lda, n
!      .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: a(:, :), work(:)
      Real (Kind=nag_wp), Allocatable  :: rwork(:)
      Integer, Allocatable         :: ipiv(:)
!      .. Executable Statements ..
      Write (nout,*) 'F07AUF Example Program Results'
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) n
      lda = n
      Allocate (a(lda,n),work(2*n),rwork(2*n),ipiv(n))

!      Read A from data file

      Read (nin,*)(a(i,1:n),i=1,n)

!      Compute norm of A
!      f06uaf is the NAG name equivalent of the LAPACK auxiliary zlange
      anorm = zlange(norm,n,n,a,lda,rwork)

!      Factorize A
!      The NAG name equivalent of zgetrf is f07arf

```

```

Call zgetrf(n,n,a,lda,ipiv,info)

Write (nout,*)
If (info==0) Then

!      Estimate condition number

!      The NAG name equivalent of zgecon is f07auf
Call zgecon(norm,n,a,lda,anorm,rcond,work,rwork,info)

If (rcond>=x02ajf()) Then
  Write (nout,99999) 'Estimate of condition number =',      &
    1.0_nag_wp/rcond
Else
  Write (nout,*) 'A is singular to working precision'
End If
Else
  Write (nout,*) 'The factor U is singular'
End If

99999 Format (1X,A,1P,E10.2)
End Program f07aufe

```

10.2 Program Data

```

F07AUF Example Program Data
  4                                     :Value of N
(-1.34, 2.55) ( 0.28, 3.17) (-6.39,-2.20) ( 0.72,-0.92)
(-0.17,-1.41) ( 3.31,-0.15) (-0.15, 1.34) ( 1.29, 1.38)
(-3.29,-2.39) (-1.91, 4.42) (-0.14,-1.35) ( 1.72, 1.35)
( 2.41, 0.39) (-0.56, 1.47) (-0.83,-0.69) (-1.96, 0.67) :End of matrix A

```

10.3 Program Results

```

F07AUF Example Program Results

Estimate of condition number = 1.50E+02

```
